
¿QUE NECESIDAD TIENEN?
Tenemos que realizar un pulsador inalámbrico para que cuando un usuario de Aspace necesite ayuda en el aula de informática, acuda un cuidador.
¿CON QUE VAMOS A TRABAJAR?
Vamos a trabajar con un Microcontrolador. El ESP32 es un microcontrolador potente y versátil que ofrece una serie de características, incluyendo conectividad WiFi y Bluetooth, procesamiento de doble núcleo, y una gama de periféricos. El ESP32 es ideal para proyectos IoT y se puede utilizar para crear una amplia gama de dispositivos inteligentes. 

Este es dispositivo que vamos a hacer inteligente para que haga la función o el proceso que nosotros queramos. Como podéis ver en el esquema tiene un montón de pines, y cada uno de ellos tiene una función o varias, por tanto, nosotros utilizaremos algunos de esos pines para poder realizar las conexiones de los elementos que utilizaremos para que pueda dar el aviso al cuidador desde el aula de informática, y otro dispositivo con otros pines y otros componentes en donde se encuentra el cuidador.
¿COMO TRANSMITIMOS EL AVISO DEL MENSAJE?
Como comentamos al principio trabajamos con el ESP32 es un microcontrolador que tiene conectividad WiFi y Bluetooth ideal para proyectos IoT. 
¿Qué ES IoT?
 La Internet de las cosas (IoT) describe la red de objetos físicos ("cosas") que llevan incorporados sensores, software y otras tecnologías con el fin de conectarse e intercambiar datos con otros dispositivos y sistemas a través de Internet.
Servidor Mosquito (MQTT)
MQTT es un protocolo de mensajería basado un conjunto de reglas, que se utiliza para la comunicación de un equipo a otro. Los sensores inteligentes, los dispositivos portátiles y otros dispositivos de Internet de las cosas (IoT) generalmente tienen que transmitir y recibir datos a través de una red con recursos restringidos y un ancho de banda limitado. Estos dispositivos IoT utilizan MQTT para la transmisión de datos, ya que resulta fácil de implementar y puede comunicar datos IoT de manera eficiente. MQTT admite la mensajería entre dispositivos a la nube y la nube al dispositivo.
¿Cómo conecto yo múltiples sensores a una red troncal para consolidar datos, sin tener que empeñar un riñón?

El contenido generado por IA puede ser incorrecto.]Este problema viene de atrás y la solución pasa por los servidores o Broker de MQTT. Estas siglas representan “Message Queing Transport Protocol” (Protocolo de transporte de colas de mensajes) y son el resultado de un problema muy similar al que comentamos más arriba, que tuvo gente de IBM a finales de los 90 y que ha devenido en un estándar abierto que podemos montar con cierta facilidad en nuestros PC Windows O Linux.
MQTT es un estándar de encapsulado de mensajes, de forma que clientes como ESP32 y similares puedan enviar información a un servidor central que gestiona esa información de forma organizada, de modo que hay usuarios que publican información y otros que suscriben información.
Normalmente los sensores publicarán la información que ellos leen de acuerdo con un cierto tópico “Temp” (como en nuestro caso la detección de una llamada) o “Laboratorio/Temp” y si alguien suscribe ese tópico (en nuestro caso el reproductor mp3 que emitirá el mensaje grabado), recibirá una copia de la información correspondiente cada vez que un cliente publique cualquier cosa en ese tópico.
¿Complicado? No tanto. Solo es confuso al principio. Imagínate como en nuestro caso que tenemos 2 ESP32 (uno con el sensor de proximidad y el otro que emite el mensaje que ennuestro caso está en otra sala). Cada uno de tus ESP32 puede conectarse a un servidor MQTT a través de WIFI y cada vez se produce una llamada, publica un mensaje en el Broker en la categoría “Aspace/Llamada”. El servidor va recibiendo esa información y comprueba en su lista de interesados (Aquellos que le dijeron que querían suscribir el tópico  “Aspace/ Llamada”, en nuestro caso el otro ESP32 que tiene el reproductor) una copia del valor recibido de forma automática sin necesidad de intervención manual.
El sistema es de los más limpio. Si sabes que publican los clientes puedes suscribirte al tema y vas recibiendo la información a medida que se produce. Además, el bróker MQTT puede manejar cientos de publicaciones y encargarse de repartir el juego de forma sencilla para los clientes que publican y los clientes que suscriben solo lo que les interesa.
Es importante entender que el bróker MQTT no puede almacenar información porque no es una base de datos.
Por eso en esta parte, vamos a ver que es bastante sencillo instalar un bróker MQTT en Windows y en Raspberry pi o Linux e iremos viendo como instalar todo lo necesario para tener nuestro propio servidor MQTT. 

Para saber cómo instalamos Mosquito en una Raspberry Pi, podemos seguir el siguiente enlace:
https://www.luisllamas.es/como-instalar-mosquitto-mqtt-en-raspberry-pi/
QUE UTILIZO COMO PULSADOR INALAMBRICO?
Como hemos comentado, utilizamos un ESP32 y un dispositivo que se basa en la orientación de los murciélagos (sensor ultrasónico), el dispositivo se denomina HC-SR04.
¿Cómo funciona el sensor ultrasónico HC-SR04?
El sensor ultrasónico utiliza un sonar para determinar la distancia a un objeto. Así es como funciona:
El transmisor de ultrasonidos (pin disparador) emite un sonido de alta frecuencia (40 kHz).
El sonido viaja por el aire. Si encuentra un objeto, rebota hacia el módulo.
El receptor de ultrasonidos (eco pin) recibe el sonido reflejado (eco).

DIAGRAMA DE BLOQUES
El diagrama de bloques representa que es lo que tenemos que programar en el ESP32, así el primer rectángulo nos indica que tenemos que definir el sensor ultrasónico y los pines a los que lo conectaremos a nuestro microcontrolador, el segundo bloque lo que indica es que tenemos que medir el tiempo que tarda en regresar una onda de sonido al rebotar contra el obstáculo, y el tercero es que debemos de aplicar una fórmula para traducir ese tiempo en una distancia.
Por último tenemos la acción que es encender el led si la distancia es menor de 20cm y emitir el mensaje al servidor mosquito a través de la wifi de Aspace.

PROGRAMA
#include <WiFi.h>
#include <PubSubClient.h>
// Pines del sensor y LED
#define TRIG_PIN 5
#define ECHO_PIN 18
#define LED_PIN 2
// Configuración WiFi
const char* ssid = "ASPACE_coruna";
const char* password = "AnuRoc/EcapSA.77";
// Configuración del broker MQTT
const char* mqtt_server = "192.168.30.199";  // IP del servidor Mosquitto
const int mqtt_port = 1883;
const char* mqtt_user = "";  // No requerido si Mosquitto no lo pide
const char* mqtt_password = "";
WiFiClient espClient;
PubSubClient client(espClient);
// Variables para control de obstáculo
bool obstaculoPresente = false;
unsigned long ultimaAlarma = 0;
const unsigned long tiempoEspera = 5000; // Tiempo mínimo entre alarmas (ms)
void setup_wifi() {
  delay(100);
  Serial.println();
  Serial.print("Conectando a ");
  Serial.println(ssid);
  WiFi.begin(ssid, password);
  while (WiFi.status() != WL_CONNECTED) {
    delay(500);
    Serial.print(".");
  }
  Serial.println("\nWiFi conectado");
  Serial.print("Dirección IP: ");
  Serial.println(WiFi.localIP());
}
void reconnect() {
  while (!client.connected()) {
    Serial.print("Conectando al servidor MQTT...");
    if (client.connect("ESP32_Emisor", mqtt_user, mqtt_password)) {
      Serial.println("conectado");
    } else {
      Serial.print("Fallo, rc=");
      Serial.print(client.state());
      Serial.println(" intentando de nuevo en 5 segundos");
      delay(5000);
    }
  }
}
void setup() {
  Serial.begin(115200);
  pinMode(TRIG_PIN, OUTPUT);
  pinMode(ECHO_PIN, INPUT);
  pinMode(LED_PIN, OUTPUT);
  setup_wifi();
  client.setServer(mqtt_server, mqtt_port);
}
void loop() {
  if (!client.connected()) {
    reconnect();
  }
  client.loop();
  long duration;
  float distance;
  // Medición de distancia
  digitalWrite(TRIG_PIN, LOW);
  delayMicroseconds(2);
  digitalWrite(TRIG_PIN, HIGH);
  delayMicroseconds(10);
  digitalWrite(TRIG_PIN, LOW);
  duration = pulseIn(ECHO_PIN, HIGH, 20000); // timeout de 20 ms
  distance = (duration * 0.0343) / 2;
  Serial.print("Distancia: ");
  Serial.print(distance);
  Serial.println(" cm");

  unsigned long ahora = millis();
  if (distance > 0 && distance <= 20.0) {
    digitalWrite(LED_PIN, HIGH);
    if (!obstaculoPresente && ahora - ultimaAlarma > tiempoEspera) {
      Serial.println("Obstáculo nuevo detectado, enviando alarma");
      client.publish("alarma/obstaculo", "Obstáculo detectado");
      obstaculoPresente = true;
      ultimaAlarma = ahora;
    }
  } else {
    digitalWrite(LED_PIN, LOW);
    obstaculoPresente = false;
  }
  delay(200);
}
ESQUEMA DE MONTAJE Y MATERIALES:
Como comentamos antes, para la realización de el pulsador de aviso, necesitamos los siguientes materiales:
· ESP32
· Diodo Led Rojo
· Resistencias: 220Ω, 1kΩ, 2kΩ
· sensor ultrasónico HC-SR04

EL AVISADOR (RECEPTOR)  DE LLAMADA
Como lo que recibimos es un mensaje de audio, grabado por un usuario de Aspace, pidiendo que lo ayuden en el aula de informática, en el caso del receptor, a parte del ESP32 tendremos que incluir un elemento que nos permita oír el mensaje, esto lo haremos con un altavoz, y un reproductor de mp3. Como veremos en las características del mp3 este puede conectar un altavoz directamente, pero la potencia de emisión del mensaje, tras haber realizado pruebas no es la suficiente, por lo tanto, temos que amplificar la señal y esto lo haremos a través de un módulo amplificador PAM8403. Las características de estos dos nuevos componentes son las siguientes:
DFPlayer mini
Es un dispositivo válido para ESP32 que incluye un lector de tarjetas SD, en la que podemos almacenar el mensaje que queremos reproducir, en nuestro caso el aviso de que alguien acuda al aula de informática, también incluye la electrónica necesaria para mover un pequeño altavoz o una salida de línea, por no mencionar la capacidad de leer los mp3 de un USB Flash y varios botones que nos permita controlar el mp3 a reproducir, así como el volumen. 
Básicamente, podemos usar un par de altavoces estéreo de hasta 3 Watios en los pines SPK_1 y SPK2, o una salida de línea estéreo o auriculares, en los pines DAC_L y DAC_R. A su vez podemos leer los mp3 desde la tarjeta SD o bien desde un USB conectado a los pines USB+ y USB-
También tenemos una puerta serie para controlar el dispositivo mediante mensaje y varios pines en los que podemos conectar botones para controlar la reproducción.

Amplificador PAM8403
Este módulo amplificador está basado en el chip PAM8403, un amplificador conmutado de clase D y con dos canales de sonido. Cada canal es capaz de proporcionar 3W de potencia máxima, cuando se alimenta a 5V y los altavoces son de 4Ohm.
El uso es realmente sencillo, solo hay que alimentarlo con entre 2,5 y 5,5V. Al ser un amplificador conmutado, el consumo es muy reducido y es perfecto para usarlo con baterías. Con un volumen razonable, el consumo no sube de 0,5A. Perfecto para ser alimentado desde USB.

ESQUEMA Y COMPONENTES
Para la realización del avisador (receptor) de llamada utilizamos los siguientes componentes:
· ESP32
· Pulsador NA
· Led Rojo
· Resistencia 220Ω
· DFPlayerMini
· Amplificador PAM8403
· Altavoz 6W 8 Ω

DIAGRAMA DE FLUJO

PROGRAMA
#include <WiFi.h>
#include <PubSubClient.h>
#include <HardwareSerial.h>
#include <DFRobotDFPlayerMini.h>
// Pines
#define LED_PIN 2
#define RXD2 16
#define TXD2 17
#define BUTTON_PIN 4
// WiFi
const char* ssid = "ASPACE_coruna";
const char* password = "AnuRoc/EcapSA.77";
// MQTT
const char* mqtt_server = "192.168.30.199";
const int mqtt_port = 1883;
const char* mqtt_user = "";           // Sin usuario
const char* mqtt_password = "";       // Sin contraseña
const char* mqtt_topic_sub = "alarma/obstaculo";
const char* mqtt_topic_pub = "alarma/confirmacion";

WiFiClient espClient;
PubSubClient client(espClient);
// DFPlayer
HardwareSerial mySerial(2);
DFRobotDFPlayerMini myDFPlayer;
bool alarmaActiva = false;
bool reproduciendo = false;
unsigned long tiempoInicioReproduccion = 0;
const unsigned long duracionMP3 = 10000; // Duración estimada del audio en ms
void callback(char* topic, byte* payload, unsigned int length) {
  String message;
  for (unsigned int i = 0; i < length; i++) {
    message += (char)payload[i];
  }
  if (String(topic) == mqtt_topic_sub && message == "Obstáculo detectado") {
    Serial.println("Mensaje MQTT recibido: Obstáculo detectado");
    alarmaActiva = true;
    reproduciendo = false;
  }
}
void setup_wifi() {
  delay(100);
  Serial.println("Conectando a WiFi...");
  WiFi.begin(ssid, password);
  while (WiFi.status() != WL_CONNECTED) {
    delay(500);
    Serial.print(".");
  }
  Serial.println("\nWiFi conectado");
  Serial.print("IP: ");
  Serial.println(WiFi.localIP());
}
void reconnect() {
  while (!client.connected()) {
    Serial.print("Conectando al broker MQTT...");
    if (client.connect("ESP32_DFPlayer")) {  // Sin usuario ni contraseña
      Serial.println("Conectado");
      client.subscribe(mqtt_topic_sub);
    } else {
      Serial.print("Fallo, rc=");
      Serial.print(client.state());
      Serial.println(". Reintentando en 5 segundos...");
      delay(5000);
    }
  }
}
void setup() {
  Serial.begin(115200);
  pinMode(LED_PIN, OUTPUT);
  pinMode(BUTTON_PIN, INPUT_PULLUP);
  digitalWrite(LED_PIN, LOW);
  setup_wifi();
  client.setServer(mqtt_server, mqtt_port);
  client.setCallback(callback);
  mySerial.begin(9600, SERIAL_8N1, RXD2, TXD2);
  if (!myDFPlayer.begin(mySerial)) {
    Serial.println("Error al iniciar DFPlayer.");
    while (true);
  }
  myDFPlayer.volume(25);  // Ajusta según necesidad
  Serial.println("DFPlayer listo");
}
void loop() {
  if (!client.connected()) {
    reconnect();
  }
  client.loop();
  if (alarmaActiva && digitalRead(BUTTON_PIN) == LOW) {
    alarmaActiva = false;
    reproduciendo = false;
    digitalWrite(LED_PIN, LOW);
    Serial.println("Alarma aceptada por el usuario");
    client.publish(mqtt_topic_pub, "ALARMA ACEPTADA");
    delay(500); // debounce
  }
  if (alarmaActiva && !reproduciendo) {
    Serial.println("Reproduciendo alarma...");
    digitalWrite(LED_PIN, HIGH);
    myDFPlayer.play(1);  // 0001.mp3
    reproduciendo = true;
    tiempoInicioReproduccion = millis();
  }
  if (reproduciendo && (millis() - tiempoInicioReproduccion >= duracionMP3)) {
    digitalWrite(LED_PIN, LOW);
    reproduciendo = false;
  }
}

Aqui esta la distribucién de pines del sensor ultrasénico HC-SR04.

